
1 Newton-Verfahren und automatische Differentiation
1.1 Banachscher Fixpunktsatz
Gegeben sei ein vollständiger metrischer Raum (𝑋, 𝑑(., .) und eine Abbildung Φ∶ 𝑋 → 𝑋 auf diesem Raum.

Weiterhin gelte folgendes:

• Es gibt ein abgeschlossenes Gebiet 𝑀 ⊆ 𝑋 (zum Beispiel 𝑀 = 𝑋), aus dem Φ nicht herausführt: Φ(𝑀) ⊆
𝑀 und

• Die Abbildung Φ ist auf 𝑀 kontrahierend, d.h., es gibt ein 0 ≤ 𝑘 < 1 so dass für alle 𝑥, 𝑦 ∈ 𝑀 gilt

𝑑(Φ(𝑥), Φ(𝑦)) ≤ 𝑘 ⋅ 𝑑(𝑥, 𝑦)

Dann gilt: 1. Φ hat in 𝑀 genau einen Fixpunkt 𝑥∗ so dass

Φ(𝑥∗) = 𝑥∗

2. Sei eine Folge (𝑥𝑖)𝑖∈ℕ iterativ definiert durch 𝑥𝑖+1 = Φ(𝑥𝑖) mit einem beliebigen Startwert 𝑥0 ∈ 𝑀 . Dann ist
diese Folge konvergent und sie konvergiert gegen den Fixpunkt:

lim
𝑖→∞

𝑥𝑖 = 𝑥∗.

1.2 Newton-Verfahren zur Nullstellensuche einer Funktion 𝑓(𝑥)

𝑥𝑖+1 = Φ(𝑥𝑖) ∶= 𝑥𝑖 − 𝑓(𝑥𝑖)
𝑓 ′(𝑥𝑖)

Hinreichende Bedingung:

• Sei 𝑓(𝑥) mindestens einmal stetig differenzierbar.
• Sei an einer Nullstelle 𝑓(𝑥∗) = 0 die erste Ableitung nicht Null: 𝑓 ′(𝑥∗) ≠ 0.

Dann gibt es ein 𝛿 > 0, so dass Φ in dem Gebiet 𝑀 = [𝑥∗ − 𝛿, 𝑥∗ + 𝛿] kontrahierend ist und daher das
Newtonverfahren mit einem Startwert aus diesem Gebiet konvergiert.

1.3 Newton-Verfahren zur Nullstellensuche eines nichtlinearen Gleichungssystems
Gegeben sei das (i.A. nichtlineare) Gleichungssystem von 𝑛 Gleichungen im R𝑛:

𝑓1(𝑥1, 𝑥2, … , 𝑥𝑛) = 0
𝑓2(𝑥1, 𝑥2, … , 𝑥𝑛) = 0

⋮ oder kurz: ⃗𝑓( ⃗𝑥) = 0
𝑓𝑛(𝑥1, 𝑥2, … , 𝑥𝑛) = 0

In diesem Fall verallgemeinert sich die Iterationsvorschrift zu

⃗𝑥𝑖+1 = ⃗𝑥𝑖 − J−1 ⋅ ⃗𝑓 ∣𝑥𝑖

wobei J die Jacobi-Matrix der partiellen ersten Ableitungen

J = 𝜕(𝑓1, .., 𝑓𝑚)
𝜕(𝑥1, .., 𝑥𝑛) =

⎡
⎢⎢⎢
⎣

𝜕𝑓1
𝜕𝑥1

⋯ 𝜕𝑓1
𝜕𝑥𝑛

⋮ ⋱ ⋮
𝜕𝑓𝑚
𝜕𝑥1

⋯ 𝜕𝑓𝑚
𝜕𝑥𝑛

⎤
⎥⎥⎥
⎦

bezeichnet.

In der numerischen Implementierung berechnet man natürlich kein Matrixinverses (langsam und instabil),
sondern löst das Gleichungssystem J ⃗𝑧 = ⃗𝑓 mit anschliessendem ⃗𝑥𝑖+1 = ⃗𝑥𝑖 − ⃗𝑧.

1.4 Automatisches Differenzieren
1.4.1 Duale Zahlen

Wir erweitern den Körper R zum Ring D der dualen Zahlen (Clifford 1873), indem wir ein Element 𝜖 hinzufügen
mit der Eigenschaft 𝜖2 = 0.

Das Prinzip ist dasselbe wie bei der Erweiterung R ⇒ C durch Hinzufügen des Elements 𝑖 mit der
Eigenschaft 𝑖2 = −1. Allerdings ist 𝜖 ein Nullteiler und damit D kein Körper, sondern “nur” ein
Ring.

Die Arithmetik ist offensichtlich: Sei 𝑧1 = 𝑎 + 𝑏𝜖 und 𝑧2 = 𝑎′ + 𝑏′𝜖 mit 𝑎, 𝑏, 𝑎′, 𝑏′ ∈ R. Dann ist

𝑧1 + 𝑧2 = (𝑎 + 𝑎′) + (𝑏 + 𝑏′)𝜖
𝑧1 ⋅ 𝑧2 = (𝑎 + 𝑏𝜖)(𝑎′ + 𝑏′𝜖) = 𝑎𝑎′ + (𝑎𝑏′ + 𝑎′𝑏)𝜖 da 𝜖2 = 0

𝑧1
𝑧2

= 𝑎 + 𝑏𝜖
𝑎′ + 𝑏′𝜖 = (𝑎 + 𝑏𝜖)(𝑎′ − 𝑏′𝜖)

(𝑎′ + 𝑏′𝜖)(𝑎′ − 𝑏′𝜖) = 𝑎𝑎′ + (𝑎′𝑏 − 𝑎𝑏′)𝜖
𝑎′2 = 𝑎

𝑎′ + 𝑎′𝑏 − 𝑎𝑏′

𝑎′2 𝜖

usw.

Mit dem Binomischen Satz wird
(𝑎 + 𝑏𝜖)𝑛 = 𝑎𝑛 + 𝑛𝑎𝑛−1𝑏𝜖

da alle höheren Potenzen von 𝜖 verschwinden.

Also gilt für jedes Polynom
𝑃(𝑎 + 𝑏𝜖) = 𝑃(𝑎) + 𝑃 ′(𝑎)𝑏𝜖

und insbesondere
𝑃(𝑥 + 𝜖) = 𝑃(𝑥) + 𝑃 ′(𝑥)𝜖

Weiterhin bricht jede Taylorentwicklung einer Funktion 𝑓(𝑥 + 𝜖) um die Stelle 𝑥 nach 2 Termen ab:

𝑓(𝑥 + 𝜖) = 𝑓(𝑥) + 𝑓 ′(𝑥)𝜖

Das ist (eine Implementation der) automatischen Differentation:

Wenn wir die Arithmetik dualer Zahlen so implementieren, dass wir für eine Funktion 𝑓(𝑥) z.B. den
Wert 𝑓(17 + 1𝜖) als duale Zahl 𝑎 + 𝑏𝜖 ausrechnen können, dann ist 𝑎 = 𝑓(17) und 𝑏 = 𝑓 ′(17). Wir
haben den Funktionswert und Wert der ersten Ableitung berechnet!

[1]: # Dual numbers, invented by Clifford in 1873. \bbiD

struct ⅅ <: Number
a :: Float64
b :: Float64

end

[2]: import Base: +,-,*,/

z1::ⅅ + z2::ⅅ = ⅅ(z1.a + z2.a, z1.b + z2.b)
z1::ⅅ - z2::ⅅ = ⅅ(z1.a - z2.a, z1.b - z2.b)
z1::ⅅ * z2::ⅅ = ⅅ(z1.a * z2.a, z1.a * z2.b + z1.b * z2.a)
z1::ⅅ / z2::ⅅ = ⅅ(z1.a / z2.a, (z2.a * z1.b - z1.a * z2.b)/z2.a^2)

# Regeln für elementare Funktionen, kleine Auswahl
import Base: sin, log
sin(z::ⅅ) = ⅅ(sin(z.a), cos(z.a) * z.b)
log(z::ⅅ) = ⅅ(log(z.a), 1/z.a * z.b)

Base.convert(::Type{ⅅ}, x::Real) = ⅅ(x, zero(x))
Base.promote_rule(::Type{ⅅ}, ::Type{<:Number}) = ⅅ

[3]: # Der Ableitungsoperator: berechnet den Wert f'(x) an der Stelle x

∂(f, x) = f(ⅅ(x, 1)).b

[3]: ∂ (generic function with 1 method)

[4]: # Zum testen

f(x) = 4x^2 + 2x + 5sin(3x)

# Ableitung
fp(x) = 8x + 2 + 15cos(3x)

[4]: fp (generic function with 1 method)

[5]: @show f(ⅅ(5,1)) f(5) fp(5);

f(ⅅ(5, 1)) = ⅅ(113.25143920078558, 30.60468130711768)
f(5) = 113.25143920078558
fp(5) = 30.60468130711768

[6]: ∂(f,5)

[6]: 30.60468130711768

[7]: # Implementation von 1dim Newton,
# f' wird durch Arithmetik dualer Zahlen berechnet

function Newton1(fct, x0; MaxIter = 50, Eps = 1.e-14)
xold = x0
i = 0
while true

i += 1
if i > MaxIter

println(stderr, "Max number of iterations reached without convergence, stop")
return

end
f = fct( ⅅ(xold, 1) ) # Das liefert f(xold) und f'(xold)

xnew = xold - f.a/f.b
println(i," ", xnew)
if abs(xnew-xold) < Eps

return xnew
end
xold = xnew

end
end

[7]: Newton1 (generic function with 1 method)

[8]: Newton1(x-> 2-x^2, 1)

1 1.5
2 1.4166666666666667
3 1.4142156862745099
4 1.4142135623746899
5 1.4142135623730951
6 1.414213562373095

[8]: 1.414213562373095

[9]: x = Newton1(x->2-x^3, 2) # 3. Wurzel 2

1 1.5
2 1.2962962962962963
3 1.2609322247417485
4 1.2599218605659261
5 1.2599210498953948
6 1.2599210498948732
7 1.2599210498948732

[9]: 1.2599210498948732

[10]: x^3

[10]: 2.0

1.5 Newton-Verfahren in der Optimierung
• Wir suchen Maximum/Minimum von 𝐹(𝑥1, 𝑥2, ..., 𝑥𝑛) im ℝ𝑛.

• Wir suchen dazu die gemeinsame Nullstelle der 𝑛 Gleichungen

𝑓1 = 𝜕𝐹( ⃗𝑥)
𝜕𝑥1

= 0, 𝑓2 = 𝜕𝐹( ⃗𝑥)
𝜕𝑥2

= 0, … , 𝑓𝑛 = 𝜕𝐹( ⃗𝑥)
𝜕𝑥𝑛

= 0

• Damit wird der Jacobian 𝐽 = 𝜕(𝑓1, .., 𝑓𝑚)
𝜕(𝑥1, .., 𝑥𝑛) der ersten Ableitungen der 𝑓𝑖 zur Hesse-Matrix (Hessian) der

zweiten Ableitungen von 𝐹 :

H = 𝜕2𝐹
𝜕𝑥𝑖 𝜕𝑥𝑗

=

⎡
⎢
⎢
⎢
⎢
⎢
⎢
⎢
⎣

𝜕2𝐹
𝜕𝑥2

1

𝜕2𝐹
𝜕𝑥1 𝜕𝑥2

⋯ 𝜕2𝐹
𝜕𝑥1 𝜕𝑥𝑛

𝜕2𝐹
𝜕𝑥2 𝜕𝑥1

𝜕2𝐹
𝜕𝑥2

2
⋯ 𝜕2𝐹

𝜕𝑥2 𝜕𝑥𝑛
⋮ ⋮ ⋱ ⋮

𝜕2𝐹
𝜕𝑥𝑛 𝜕𝑥1

𝜕2𝐹
𝜕𝑥𝑛 𝜕𝑥2

⋯ 𝜕2𝐹
𝜕𝑥2𝑛

⎤
⎥
⎥
⎥
⎥
⎥
⎥
⎥
⎦

Die Iterationsvorschrift wird damit zu

⃗𝑥𝑛𝑒𝑤 = ⃗𝑥𝑜𝑙𝑑 − H−1 ⋅ grad𝐹 ∣𝑥𝑜𝑙𝑑

1.6 Das Paket ForwardDiff.jl
• implementiert automatische Differentiation auch für Funktionen mehrerer Variablen
• Funktionen:

ForwardDiff.derivative()
ForwardDiff.gradient()
ForwardDiff.Jacobian()
ForwardDiff.hessian()
...

• Funktionen mehrere Variablen müssen als Funktionen von Vektoren angelegt sein.

[11]: using ForwardDiff, # hessian, gradient
LinearAlgebra, # norm of arrays
Plots

[12]: """
Mehrdimensionale Extremwertsuche mittels Newton-Verfahren,
F: Funktion eines n-vektors
x0vec: n-Vektor der Startwerte
MaxIter: max. Anzahl von Iterationen
Eps: Ziel erreicht, wenn Verbesserung in der Vektornorm ||x_neu-x_alt|| < Eps

Zur späteren Visualisierung werden die m Vektoren
(m = Anzahl der benötigten Iterationsschritte)
als nxm-Matrix zurückgegeben.
"""
function NewtonOpti(F, x0vec; MaxIter = 50, Eps = 1.e-14)

Xs = copy(x0vec)
xold = x0vec
i = 0
while true

i += 1
if i > MaxIter

println(stderr, "Max number of iterations reached without convergence, stop.")
return

end
H = ForwardDiff.hessian(F, xold)
del = H \ ForwardDiff.gradient(F, xold)
xnew = xold - del
Xs = hcat(Xs, xnew)
println(xnew)
if norm(del) < Eps

println("Extremum reached after $i iterations at $xnew")
return Xs

end
xold = xnew

end
end

[12]: NewtonOpti

[13]: # unsere 2-dimensionale Testfunktion

function f(x,y)
z = (1. - x/2 + x^5 + y^3) * exp(-x^2 - y^2)
return z

end

# und jetzt mit Vektor-Argument, wie es in ForwardDiff
# und damit in NewtonOpti() gebraucht wird:

function F(xvec)
return f(xvec[1], xvec[2])

end

[13]: F (generic function with 1 method)

[14]: # so sieht sie aus:

surface( -3:0.02:3, -3:0.02:3, f)
[14]:

[15]: # im Contourplot sieht man die Minima und Maxima

contour( -3:0.02:3, -3:0.02:3, f, fill=true, c=:Blues, levels=20, contour_labels=false)
[15]:

[16]: # und los gehts:

xs = NewtonOpti(F, [2.,0.1])

[0.1332054959300537, -0.290375646177925]
[-0.5092885621074995, 0.08142783320806135]
[-0.24479556529773994, -0.03277048936762038]
[-0.22026161325055604, -0.0011927350354439467]
[-0.22004307520481325, -1.91323696826411e-6]
[-0.22004305442098376, -4.94875303235653e-12]
[-0.2200430544209836, -3.310822713486171e-23]
[-0.22004305442098357, 0.0]
Extremum reached after 8 iterations at [-0.22004305442098357, 0.0]

[16]: 2×9 Matrix{Float64}:
2.0 0.133205 -0.509289 -0.244796 … -0.220043 -0.220043
0.1 -0.290376 0.0814278 -0.0327705 -3.31082e-23 0.0

[17]: # .. wir zeichen die Schritte des Nowton-verfahrens hin zum
# gefundenen Maximum ein:

plot!(xs[1,:], xs[2,:], marker=:circle, markersize=5)
[17]:

[18]: # wir kombinieren gleich Berechnung und Visualisierung
# und bauen die Möglichkeit ein,
# Ausschnitte mit den Grenzen [(x1,x2),(y1,y2)] zu plotten

function MakePlot(x0; x1=-3, x2=3, y1=-3, y2=3)
xs = NewtonOpti(F, x0)
p = contour( range(x1,x2,300), range(y1,y2,300), f, fill=true, c=:Blues, levels=20,

contour_labels=false)
plot!(xs[1,:], xs[2,:], marker=:circle, markersize=5, linesize=2)

end

[18]: MakePlot (generic function with 1 method)

[19]: # und das obige Bild nochmal im Ausschnitt
# ... Interesanterweise geht der erste Iterationsschritt am nächstgelegenem
# Maximum vorbei und das Verfahren läuft zum anderen Maximum...

MakePlot([2,.1], x1=-1.3, x2=2.5, y1=-.5, y2=.3)

[0.1332054959300537, -0.290375646177925]
[-0.5092885621074995, 0.08142783320806135]
[-0.24479556529773994, -0.03277048936762038]
[-0.22026161325055604, -0.0011927350354439467]
[-0.22004307520481325, -1.91323696826411e-6]
[-0.22004305442098376, -4.94875303235653e-12]
[-0.2200430544209836, -3.310822713486171e-23]
[-0.22004305442098357, 0.0]
Extremum reached after 8 iterations at [-0.22004305442098357, 0.0]

[19]:

[20]: MakePlot([1,0.1])

[0.6956106547605012, 0.03171439130151904]
[0.7920383277477394, -0.0004734597583571504]
[0.7890525789862474, -3.9117647761667627e-7]
[0.7890542734775248, -2.518659773261041e-13]
[0.7890542734780247, -1.044578056257455e-25]
[0.7890542734780247, 0.0]
Extremum reached after 6 iterations at [0.7890542734780247, 0.0]

[20]:

[21]: # ..und diesmal treffen wir den Saddelpunkt.
# da verschwinden natürlich auch alle partiellen Ableitungen

MakePlot([1,0.1], x1=0.5, x2=1.2, y1=-.2, y2=.2)

[0.6956106547605012, 0.03171439130151904]
[0.7920383277477394, -0.0004734597583571504]
[0.7890525789862474, -3.9117647761667627e-7]
[0.7890542734775248, -2.518659773261041e-13]
[0.7890542734780247, -1.044578056257455e-25]
[0.7890542734780247, 0.0]
Extremum reached after 6 iterations at [0.7890542734780247, 0.0]

[21]:

[22]: # mit diesem Startwert geht's zu einem der Minima

MakePlot( [-1.8,0.2])

[-1.6584806215974253, -0.03160877603265452]
[-1.6881945075073073, 0.0003759256128861993]
[-1.6888385202356129, 1.696332552631732e-8]
[-1.688838885976237, 3.627747814500337e-17]
[-1.6888388859763552, 0.0]
[-1.6888388859763555, 0.0]
Extremum reached after 6 iterations at [-1.6888388859763555, 0.0]

[22]:

[23]: #nochmal als Ausschnitt

MakePlot([-1.8,.2], x1=-2, x2=-1.4, y1=-.4, y2=.3)

[-1.6584806215974253, -0.03160877603265452]
[-1.6881945075073073, 0.0003759256128861993]
[-1.6888385202356129, 1.696332552631732e-8]
[-1.688838885976237, 3.627747814500337e-17]
[-1.6888388859763552, 0.0]
[-1.6888388859763555, 0.0]
Extremum reached after 6 iterations at [-1.6888388859763555, 0.0]

[23]:

[24]: # und mit diesem Startwert erreichen wir auch das rechte Maximum

MakePlot([1.3,0.2])

[1.7495598911691388, -0.1481347995568894]
[1.4979885189593156, 0.03284641414014772]
[1.5486102113661622, -0.0006009360354399529]
[1.5477195717036256, -5.706673652540722e-8]
[1.5477195171353288, -5.36382925634516e-16]
[1.5477195171353284, -9.860761315262648e-32]
Extremum reached after 6 iterations at [1.5477195171353284,
-9.860761315262648e-32]

[24]:

[25]: # nochmal als Ausschnitt

MakePlot([1.3,.2], x1=1.1, x2=2, y1=-.2, y2=.3)

[1.7495598911691388, -0.1481347995568894]
[1.4979885189593156, 0.03284641414014772]
[1.5486102113661622, -0.0006009360354399529]
[1.5477195717036256, -5.706673652540722e-8]
[1.5477195171353288, -5.36382925634516e-16]
[1.5477195171353284, -9.860761315262648e-32]
Extremum reached after 6 iterations at [1.5477195171353284,
-9.860761315262648e-32]

[25]:

